Name ____________________________ Class Time _____

Math 2250 Maple Project 7: Laplace Applications
F2009

Due date: See the internet due dates. Maple lab 7 has five problems L7.1, L7.2, L7.3, L7.4, L7.5.

References: Code in `maple` appears in `2250mapleL7-F2009.txt` at URL `http://www.math.utah.edu/~gustafso/`. This document: `2250mapleL7-F2009.pdf`. Other related and required documents are available at the web site.

Problem L7.1. (Periodic Wave Plots)

In the table are examples of standard periodic waves.

(a) Plot them all. Please choose an appropriate graph window for each.

(b) Piecewise expressions h are given in the table on the base interval $[0, T]$. The T-periodic extension of f off the base interval is always $h(g(t))$ where $g(t) = t - T\,\mathbf{floor}(t/T)$. Observe that $g(t)$ equals `T f2(t/T)`; see the table. Justify every maple expression in the table. In the first example, define `h1:=t->piecewise(t<1,1,t<2,-1,0);` and then plot `h1(T*f2(t/T))-f1(t)` over 3 periods [$T = 2$ for the square wave]. It should plot as the zero function.

Useful plot options are `ytickmarks=3`, `color=red`, `labels=[t,'f(t)']`, `title="square wave"`, `numpoints=100`, `thickness=2`. Combine options like this: `opts:=discont=true,thickness=2`, and use them as `plot(f(t),t=a..b,opts);`.

Maple Expression	Name	T	Piecewise Definition on $[0, T]$
`f1:=t ->(-1)^floor(t);`	square wave	2	$h_1(t) = \begin{cases} 1 & 0 \le t < 1, \\ -1 & 1 \le t < 2 \end{cases}$
`f2:=t -> t-floor(t);`	triangular wave	1	$h_2(t) = \begin{cases} t & 0 \le t < 1, \\ 0 & t = 1 \end{cases}$
`f3:=t -> 1/2+(f2(t)-1/2)*f1(t);`	sawtooth wave	2	$h_3(t) = \begin{cases} t & 0 \le t < 1, \\ 2-t & 1 \le t < 2 \end{cases}$
`f4:=t->abs(sin(t));`	rectified sine	2π	$h_4(t) = \begin{cases} \sin(t) & 0 \le t < \pi, \\ -\sin(t) & \pi \le t < 2\pi \end{cases}$
`f5:=t->sin(t)+abs(sin(t));`	half-wave rectified sine	2π	$h_5(t) = \begin{cases} \sin(t) & 0 \le t < \pi, \\ 0 & \pi \le t < 2\pi \end{cases}$
`p:= t -> (2-t)*t:` `f6:=t->p(2*f2(t/2))*f1(t/2);`	parabolic wave	4	$h_6(t) = \begin{cases} p(t) & 0 \le t < 2, \\ -p(t-2) & 2 \le t < 4 \end{cases}$
`q:=t->` `piecewise(t<Pi,sin(t),t<2*Pi,-1):` `f7:=x->q(2*Pi*f2(x/2/Pi));`	piecewise sine pulse	2π	$h_7(t) = \begin{cases} \sin(t) & 0 \le t < \pi, \\ -1 & \pi \le t < 2\pi \end{cases}$

Problem L7.2. (Hammer Hit Oscillation)

An attached mass in an undamped spring-mass system is released from rest 1 meter below the equilibrium position. After 3 seconds of oscillation, the mass is struck by a hammer with force of 5 Newtons in a downward direction.

(a) Assume the model

$$\frac{d^2x}{dt^2} + 9x = 5\delta(t-3); x(0) = 1, \frac{dx}{dt}(0) = 0,$$

where $x(t)$ denotes the displacement from equilibrium at time t and $\delta(t-3)$ denotes the Dirac delta function. Determine, using the `dsolve` example below, a piecewise-defined formula for $x(t)$. Plot $x(t)$ for $0 \le t \le 7$.

(b) Solve the following hammer-hit models `DE1` to `DE4`, given as maple expressions, using the `dsolve` example for `DE`, `IC` as a template for the solution.

(c) Express the symbolic answer for each of `DE1` to `DE4` as a piecewise-defined function. Interpret each answer physically.

```
DE:=diff(x(t),t,t)+9*x(t)=3*Dirac(t-3); IC:=x(0)=1,D(x)(0)=0;
```

```
dsolve({DE,IC},x(t),method=laplace);
# x(t) = cos(3*t)+Heaviside(t-3)*sin(-9+3*t)
convert(%,piecewise);combine(%,trig);
# x(t) = cos(3*t) for t < 3,cos(3*t)+sin(-9+3*t) for t>3, undef t=3.

DE1:=diff(x(t),t,t)+9*x(t)=5*Dirac(t-3); IC1:=x(0)=-1,D(x)(0)=1;
DE2:=diff(x(t),t,t)+9*x(t)=6*Dirac(t-3); IC2:=x(0)=1,D(x)(0)=-1;
DE3:=diff(x(t),t,t)+9*x(t)=8*Dirac(t-3); IC3:=x(0)=0,D(x)(0)=-1;
DE4:=diff(x(t),t,t)+9*x(t)=9*Dirac(t-3); IC4:=x(0)=1,D(x)(0)=0;
```

Problem L7.3. (Maple Solution of Initial Value Problems)

(a) Solve the IVP $y'' - y' - 2y = 5\sin x$, $y(0) = 1$, $y'(0) = -1$. Please use the `inttrans` package. Show the steps in Laplace's method, entirely in maple, with explicit use of maple functions `laplace(f,t,s)` and `invlaplace(F,s,t)`.

(b) Solve the pulse-input IVP

$$3y'' + 3y' + 2y = \begin{cases} 0 & \text{for} \quad t < 0, \\ 3 & \text{for} \quad 0 \le t < 4, \\ 0 & \text{for} \quad t \ge 4, \end{cases}$$

with initial data $y(0) = 0$, $y'(0) = 0$. Use any maple method. Express your answer as a piecewise-defined function.

(c) Solve the IVP $y''+y = 1+\delta(t-2\pi)$, $y(0) = 1$, $y'(0) = 0$. Use maple `dsolve`. Express the answer as a piecewise-defined function.

Problem L7.4. (Expressions for Periodic Waves)

Let h be the T-periodic extension of $f(x) = 2/10 + (7/10)\sin x + (1/10)\cos 5x$, defined on $[0, T]$, for $T = 2$.

(a) Plot $h(t)$ on the interval $[-10, 10]$. Use the composition formula $h(t) = f(g(t))$, where $g(t) = t - T\,\mathbf{floor}(t/T)$.

(b) Compute the Laplace of $h(t)$ directly from the periodic function theorem, using the sample maple code

```
int(f(g(t))*exp(-s*t),t=0..T)/(1-exp(-s*T));
```

Replacing $f(x)$ by $(1/10)\cos(5x)$ should give the answer below. The answer for $2/10 + (7/10)\sin x + (1/10)\cos 5x$ has many more terms.

$$\frac{1}{10}\,\frac{se^{2\,s} - s\cos(10) + 5\,\sin(10)}{(s^2+25)\left(-1+e^{2\,s}\right)}$$

(c) Maple finds the laplace of $g(t) = t - T\,\mathbf{floor}(t/T)$, but not the laplace of $h(t) = f(g(t))$. Express $h(t)$ as a series of pulses to get help from maple. The laplace of some terms in the series $h(t)$ can be computed, but not all. The constant term and sine term are a success but the cosine term causes maple to churn. This example shows that the periodic function theorem is a basic tool in Laplace theory. Here's the success story for this example:

```
pulse:=(t,a,b)->Heaviside(t-a)-Heaviside(t-b);
f := x -> 2/10+7/10*sin(x):  h:= t->sum(f(t-n*T)*pulse(t,n*T,n*T+T),n=0..infinity);
inttrans[laplace](h(t),t,s);
eval(%) assuming n::positive;
```

Here's what does not work. See if you can change the code and make it work, giving the answer in (b) above. Beware of testing the code below: it uses about 800mb memory and finishes with no answer.

```
pulse:=(t,a,b)->Heaviside(t-a)-Heaviside(t-b);
f := x -> (1/10)*cos(5*x):
h:= t->sum(f(t-n*T)*pulse(t,n*T,n*T+T),n=0..infinity);
inttrans[laplace](h(t),t,s);
eval(%) assuming n::positive;
```

Problem L7.5. (Resolvent Method)

The Laplace resolvent formula for the problem $\mathbf{u}' = A\mathbf{u}$, $\mathbf{u}(0) = \mathbf{u}_0$ is

$$\mathcal{L}(\mathbf{u}(t)) = (sI - A)^{-1}\mathbf{u}_0.$$

For example, $A = \begin{pmatrix} 1 & 0 \\ 0 & 2 \end{pmatrix}$ gives

$$\mathcal{L}(\mathbf{u}(t))) = \begin{pmatrix} s-1 & 0 \\ 0 & s-2 \end{pmatrix}^{-1} \mathbf{u}_0 = \begin{pmatrix} \frac{1}{s-1} & 0 \\ 0 & \frac{1}{s-2} \end{pmatrix} \mathbf{u}_0 = \begin{pmatrix} \mathcal{L}(e^t) & 0 \\ 0 & \mathcal{L}(e^{2t}) \end{pmatrix} \mathbf{u}_0,$$

which implies $\mathbf{u}(t) = \begin{pmatrix} e^t & 0 \\ 0 & e^{2t} \end{pmatrix} \mathbf{u}_0$.

The answers for the components of $\mathbf{u}$ are αe^t, βe^{2t}, according to the following maple code:

```
with(LinearAlgebra):with(inttrans):
A:=Matrix([[1,0],[0,2]]);
u0:=Vector([alpha,beta]);
B:=(s*IdentityMatrix(2)-A)^(-1).u0;
u:=Map(invlaplace,B,s,t);
```

Compute the solution $\mathbf{u}(t)$ using the resolvent formula for the following cases.

(a) $A = \begin{pmatrix} 2 & 1 \\ 1 & 2 \end{pmatrix}$, $\mathbf{u}(0) = \begin{pmatrix} 0 \\ -1 \end{pmatrix}$

(b) $A = \begin{pmatrix} 0 & 1 \\ -1 & -2 \end{pmatrix}$, $\mathbf{u}(0) = \begin{pmatrix} 1 \\ -1 \end{pmatrix}$

(c) $A = \begin{pmatrix} 2 & -5 \\ 4 & -2 \end{pmatrix}$, $\mathbf{u}(0) = \begin{pmatrix} -1 \\ 2 \end{pmatrix}$